/***************************************************************************
 *   oboe.controllers.LoadingController                                    *
 *   Copyright (C) 2011 by Felipe Manga                                    *
 *   lufeboma@gmail.com                                                    *
 *                                                                         *
 *   Permission is hereby granted, free of charge, to any person obtaining *
 *   a copy of this software and associated documentation files (the       *
 *   "Software"), to deal in the Software without restriction, including   *
 *   without limitation the rights to use, copy, modify, merge, publish,   *
 *   distribute, sublicense, and/or sell copies of the Software, and to    *
 *   permit persons to whom the Software is furnished to do so, subject to *
 *   the following conditions:                                             *
 *                                                                         *
 *   The above copyright notice and this permission notice shall be        *
 *   included in all copies or substantial portions of the Software.       *
 *                                                                         *
 *   THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,       *
 *   EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF    *
 *   MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. *
 *   IN NO EVENT SHALL THE AUTHORS BE LIABLE FOR ANY CLAIM, DAMAGES OR     *
 *   OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, *
 *   ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR *
 *   OTHER DEALINGS IN THE SOFTWARE.                                       *
 ***************************************************************************/

package oboe.controllers
{
	import flash.display.DisplayObject;
	import flash.utils.getTimer;
	import flash.utils.setTimeout;
	
	import oboe.core.Controller;
	import oboe.core.Oboe;
	import oboe.resources.Files;
	
	public class LoadingController extends Controller
	{
		private var minTime : int;
		private var startTime : int;

		public function LoadingController(time:Number, displayObject:DisplayObject=null)
		{
			super(displayObject);
			this.minTime = time;
		}
		
		protected function startLoading() : void 
		{
			var buffer : Array = getPreloadList();
			if( buffer )
			{
				Files.loadList( buffer, onLoadFirst );
			}
			else
			{
				onLoadFirst();
			}
		}
		
		private function onLoadFirst() : void 
		{
			startTime = getTimer();
			Files.loadList( getLoadList(), onLoadNormal, this );
		}
		
		private function onLoadNormal() : void
		{
			var timeSpent : int = getTimer() - startTime;
			if( timeSpent < minTime ) setTimeout( onDoneLoading, minTime - timeSpent );
			else onDoneLoading();
		}
		
		/**
		 * This function should be overridden to return a RingBuffer containing a FileRef for each file to download. 
		 * @return the RingBuffer 
		 */
		protected function getLoadList() : Array {
			return null;
		} 
		
		/**
		 * This function should be overridden to return a RingBuffer containing a FileRef for each file to download.
		 * It is called before anything is drawn on the screen. Should be used for the loading screen animations. 
		 * @return the RingBuffer 
		 */
		protected function getPreloadList() : Array {
			return null;
		}
		
		/**
		 * When everything is done loading, this function is called so that the controller can change to the next screen. 
		 * 
		 */
		protected function onDoneLoading() : void {
			// implement me
		}
	}
}